xAct`SymManipulator`

This is the doc file SymManipulatorDoc.nb of version 0.8.3 of `SymManipulator`` . Last update on 5 May 2012.

Author

Thomas Bäckdahl

Max-Planck-Institut für Gravitationsphysik

Albert Einstein Institut

Golm, Germany

`thomas.backdahl@aei.mpg.de`

Intro

`SymManipulator`` is the `xAct`` package for computations with *symmetrized tensor expressions*. By a symmetrized tensor expression we mean a sum of tensors where each term is the original expression with the indices permuted. We require the set of permutations to form a group. We call this group the *imposed symmetry group*. It is a group of signed permutations. The sign of the permutation gives the sign of the corresponding term.

Load the package

This loads the package from the default directory, for example $Home/.*Mathematica*/Applications/xAct/ for a single-user installation under Linux.

```
In[1]:= MemoryInUse[]

Out[1]= 17 148 728
```

```
In[2]:= << xAct`SymManipulator`

        ------------------------------------------------------------

        Package xAct`xPerm`  version 1.1.3, {2012, 5, 5}

        CopyRight (C) 2003-2011, Jose M.
          Martin-Garcia, under the General Public License.

        Connecting to external cygwin executable...

        Connection established.

        ------------------------------------------------------------

        Package xAct`xTensor`  version 1.0.4, {2012, 5, 5}

        CopyRight (C) 2002-2011, Jose M.
          Martin-Garcia, under the General Public License.

        ------------------------------------------------------------

        Package xAct`SymManipulator`  version 0.8.3, {2012, 5, 5}

        CopyRight (C) 2011, Thomas Bäckdahl, under the General Public License.

        ------------------------------------------------------------

        These packages come with ABSOLUTELY NO WARRANTY; for details type
          Disclaimer[]. This is free software, and you are welcome to redistribute
          it under certain conditions. See the General Public License for details.

        ------------------------------------------------------------

In[3]:= <<xAct`TexAct`

        ------------------------------------------------------------

        Package xAct`TexAct`  version 0.3.0, {2012, 5, 5}

        CopyRight (C) 2008-2012, Thomas Bäckdahl, Jose M.
          Martin-Garcia and Barry Wardell, under the General Public License.

        ------------------------------------------------------------

        These packages come with ABSOLUTELY NO WARRANTY; for details type
          Disclaimer[]. This is free software, and you are welcome to redistribute
          it under certain conditions. See the General Public License for details.

        ------------------------------------------------------------
```

Comparing, we see that what packages take in *Mathematica* 8.0:

```
In[4]:= MemoryInUse[]

Out[4]= 27 790 432
```

```
In[5]:= (Out[4] - Out[1]) / 2^20 // N

Out[5]= 10.1487
```

There are several contexts: `xAct`SymManipulator``, `xAct`xTensor``, `xAct`xPerm`` and `xAct`xCore`` contain the respective reserved words. `System`` contains *Mathematica*'s reserved words. The current context `Global`` will contain your definitions and right now is empty.

```
In[6]:= $ContextPath

Out[6]= {xAct`TexAct`, xAct`SymManipulator`, xAct`xTensor`, xAct`xPerm`,
         xAct`xCore`, PacletManager`, WebServices`, System`, Global`}

In[7]:= Context[]

Out[7]= Global`

In[8]:= ? Global`*

        Information::nomatch : No symbol matching Global`* found. >>
```

1. Example session

The package works for all dimensions. At the moment only abstract indices can be used.

Basic examples

Definition of a four dimensional manifold `M4`:

```
In[9]:= DefManifold[M4, 4, {a, b, c, d, f, p, q, r, m, l, h, j, n, t, s}]

          ** DefManifold: Defining manifold M4.

          ** DefVBundle: Defining vbundle TangentM4.
```

For most things we don’t need a metric so we do not need to define one now.

We define a tensor without symmetries:

```
In[10]:= DefTensor[T[-a, -b, -c, -d], M4]

          ** DefTensor: Defining tensor T[-a, -b, -c, -d].
```

If we want to completely symmetrize T, we can do it using just xTensor:

```
In[11]:= SymTexpr1 = ImposeSymmetry[T[-a, -b, -c, -d],
           IndexList[-a, -b, -c, -d], Symmetric[{1, 2, 3, 4}]]
```

Out[11]= $\frac{1}{24}\left(T_{abcd}+T_{abdc}+T_{acbd}+T_{acdb}+T_{adbc}+T_{adcb}+T_{bacd}+T_{badc}+T_{bcad}+T_{bcda}+T_{bdac}+T_{bdca}+T_{cabd}+T_{cadb}+T_{cbad}+T_{cbda}+T_{cdab}+T_{cdba}+T_{dabc}+T_{dacb}+T_{dbac}+T_{dbca}+T_{dcab}+T_{dcba}\right)$

This representation is often inconvenient to use so we can impose the symmetry abstractly instead using ImposeSym:

```
In[12]:= SymTexpr2 =
          ImposeSym[T[-a, -b, -c, -d], IndexList[-a, -b, -c, -d], Symmetric[{1, 2, 3, 4}]]
```

Out[12]= $\underset{(1234)}{\text{Sym}}[T]_{abcd}$

The label (1234) means that the symmetrization is taken over the slots 1, 2, 3 and 4.

We can manipulate this expression, and only expand it when we need to. We expand it with ExpandSym:

```
In[13]:= ExpandSym@SymTexpr2
```

Out[13]= $\frac{1}{24}\left(T_{abcd}+T_{abdc}+T_{acbd}+T_{acdb}+T_{adbc}+T_{adcb}+T_{bacd}+T_{badc}+T_{bcad}+T_{bcda}+T_{bdac}+T_{bdca}+T_{cabd}+T_{cadb}+T_{cbad}+T_{cbda}+T_{cdab}+T_{cdba}+T_{dabc}+T_{dacb}+T_{dbac}+T_{dbca}+T_{dcab}+T_{dcba}\right)$

We can check that this is the same as ImposeSymmetry:

```
In[14]:= Expand[% - SymTexpr1]

Out[14]= 0
```

The symmetrized expression is treated as a tensor. That is most xTensor tools will work on it. A new tensor is not defined, so all information is carried in the head of the symmetrized object.

```
In[15]:= Head@SymTexpr2

Out[15]= SymH[{T}, StrongGenSet[{1, 2, 3, 4},
           GenSet[Cycles[{1, 2}], Cycles[{2, 3}], Cycles[{3, 4}]]], (1234)]

In[16]:= xTensorQ@%

Out[16]= True

In[17]:= SymmetryGroupOfTensor@%%

Out[17]= StrongGenSet[{1, 2, 3, 4}, GenSet[Cycles[{3, 4}], Cycles[{2, 3}], Cycles[{1, 2}]]]
```

The canonicalizer can therefore work on SymH objects:

```
In[18]:= SymTexpr2 + (SymTexpr2 /. {-a -> -b, -b -> -a})
```

Out[18]= $\underset{(1234)}{\text{Sym}}[T]_{abcd} + \underset{(1234)}{\text{Sym}}[T]_{bacd}$

```
In[19]:= ToCanonical@%
```

Out[19]= $2\ \underset{(1234)}{\text{Sym}}[T]_{abcd}$

Other groups can also be imposed. For instance symmetrize over -a,-c and antisymmetrize over -b,-d:

```
In[20]:= ImposeSym[T[-a, -b, -c, -d], IndexList[-a, -b, -c, -d],
          GenSet[Cycles[{1, 3}], -Cycles[{2, 4}]]]
```

Out[20]= $\underset{(13)[24]}{\text{Sym}}[T]_{abcd}$

```
In[21]:= ExpandSym@%
```

Out[21]= $\frac{1}{4}\left(T_{abcd} - T_{adcb} + T_{cbad} - T_{cdab}\right)$

If the group has more complicaed structure, the label only shows the slots involved.

```
In[22]:= ImposeSym[T[-a, -b, -c, -d],
          IndexList[-a, -b, -c, -d], RiemannSymmetric[{1, 2, 3, 4}]]
```

Out[22]= $\underset{1234}{\text{Sym}}[T]_{abcd}$

```
In[23]:= ExpandSym@%
```

Out[23]= $\frac{1}{8}\left(T_{abcd} - T_{abdc} - T_{bacd} + T_{badc} + T_{cdab} - T_{cdba} - T_{dcab} + T_{dcba}\right)$

TexAct package can produce a nice TeX code output if the symmetrized object is possible to write in a nice way.

```
In[24]:= ImposeSym[T[-a, -b, -c, -d], IndexList[-a, -b, -c], Antisymmetric[{1, 2, 3}]]
```

Out[24]= $\underset{[123]}{\text{Sym}}[T]_{abcd}$

```
In[25]:= TexPrint@%

Out[25]= T_{[abc]d}
```

```
In[26]:= ImposeSym[T[a, -b, -c, -d], IndexList[-b, -d], Symmetric[{1, 2}]]
```

Out[26]= $\underset{(24)}{\text{Sym}}[T]^{a}{}_{bcd}$

In[27]:= **TexPrint@%**

Out[27]= T^{a}{}_{(b|c|d)}

The internal tensor can also have symmetries.

In[28]:= **DefTensor[S[-a, -b, -c, -d], M4, Symmetric[{2, 3, 4}]]**

** DefTensor: Defining tensor S[-a, -b, -c, -d].

In[29]:= **SymSexpr1 = ImposeSym[S[-a, -b, -c, -d], IndexList[-a, -b, -c, -d], Symmetric[{1, 2}]]**

Out[29]= $\underset{(12)}{\text{Sym}[S]}_{abcd}$

Now the total symmetry is a mixture of the symmetry of S and the imposed symmetry.

In[30]:= **SymmetryGroupOfTensor@Head@SymSexpr1**

Out[30]= StrongGenSet[{1, 2, 3, 4}, GenSet[Cycles[{3, 4}], Cycles[{1, 2}]]]

The calculation of this symmetry has special codes for the cases when the imposed symmetry is the symmetric or antisymmetric group on some indices. In all other cases, a more time consuming code is used to compute the mixed symmetry.

If the tensor already has the the imposed symmetry, it is automatically simplified.

In[31]:= **ImposeSym[S[-a, -b, -c, -d], IndexList[-c, -d], Symmetric[{1, 2}]]**

Out[31]= S_{abcd}

If one antisymmetrizes over a pair of symmetric slots, one will get zero. This idea have been generalized, and this is checked automatically by ImposeSym. This check works for any combination of groups.

In[32]:= **ImposeSym[S[-a, -b, -c, -d], IndexList[-c, -d], Antisymmetric[{1, 2}]]**

Out[32]= 0

In[33]:= **ImposeSym[S[-a, -b, -c, -d], IndexList[-a, -b, -c, -d], RiemannSymmetric[{1, 2, 3, 4}]]**

Out[33]= 0

Sometimes the expansion of a symmetry gives many more terms than need due to the internal symmetry.

In[34]:= **SymSexpr2 = ImposeSym[S[-a, -b, -c, -d], IndexList[-a, -b, -c, -d]]**

Out[34]= $\underset{(1234)}{\text{Sym}[S]}_{abcd}$

In[35]:= **ExpandSym@SymSexpr2**

Out[35]= $\frac{1}{24}\left(S_{abcd}+S_{abdc}+S_{acbd}+S_{acdb}+S_{adbc}+S_{adcb}+S_{bacd}+S_{badc}+S_{bcad}+S_{bcda}+S_{bdac}+S_{bdca}+S_{cabd}+S_{cadb}+S_{cbad}+S_{cbda}+S_{cdab}+S_{cdba}+S_{dabc}+S_{dacb}+S_{dbac}+S_{dbca}+S_{dcab}+S_{dcba}\right)$

In[36]:= **ToCanonical@%**

Out[36]= $\frac{1}{4}S_{abcd}+\frac{1}{4}S_{bacd}+\frac{1}{4}S_{cabd}+\frac{1}{4}S_{dabc}$

Then one can give the option SmartExpand->True. Instead of computing all elements of the imposed symmetry group, only a transversal is computed. This is a ***new feature*** in release 0.8.3 and not very well tested.

In[37]:= **ExpandSym[SymSexpr2, SmartExpand -> True]**

Out[37]= $\frac{1}{4}\left(S_{abcd}+S_{bacd}+S_{cabd}+S_{dabc}\right)$

This idea is used in the new commands SmartSymmetrize and SmartAntisymmetrize. They do the same thing as Symmetrize and Antisymmetrize, but number of terms is reduced by taking the internal symmetry into consideration. Observe that this is a ***new feature*** in release 0.8.3 and not very well tested.

In[38]:= **SmartSymmetrize[S[-a, -b, -c, -d], IndexList[-a, -b, -c, -d]]**

Out[38]= $\frac{1}{4}\left(S_{abcd}+S_{bacd}+S_{cabd}+S_{dabc}\right)$

In[39]:= **Symmetrize[S[-a, -b, -c, -d], IndexList[-a, -b, -c, -d]]**

Out[39]= $\frac{1}{24}\left(S_{abcd}+S_{abdc}+S_{acbd}+S_{acdb}+S_{adbc}+S_{adcb}+S_{bacd}+S_{badc}+S_{bcad}+S_{bcda}+S_{bdac}+S_{bdca}+S_{cabd}+S_{cadb}+S_{cbad}+S_{cbda}+S_{cdab}+S_{cdba}+S_{dabc}+S_{dacb}+S_{dbac}+S_{dbca}+S_{dcab}+S_{dcba}\right)$

In[40]:= **ToCanonical@%**

Out[40]= $\frac{1}{4}S_{abcd}+\frac{1}{4}S_{bacd}+\frac{1}{4}S_{cabd}+\frac{1}{4}S_{dabc}$

Product of tensors can be handled. In this case, interchange symmetries etc will be considered as usual.

In[41]:= **ImposeSym[T[-a, -b, -c, -d] T[-f, -h, -l, -m], IndexList[-d, -m], Symmetric[{1, 2}]]**

Out[41]= $\underset{(48)}{\mathrm{Sym}[T\,T]}_{abcdfhlm}$

```
In[42]:= SymmetryGroupOfTensor@Head@%

Out[42]= StrongGenSet[{1, 2, 3, 4, 5, 6, 7, 8},
          GenSet[Cycles[{4, 8}], Cycles[{1, 5}, {2, 6}, {3, 7}]]]

In[43]:= ImposeSym[S[-a, -b, -c, -d] T[-f, -h, -l, -m],
          IndexList[-d, -f], Symmetric[{1, 2}]]

Out[43]= Sym[S T]
         (45)      abcdfhlm
```

The order of the tensors is automatically sorted by ImposeSym.

```
In[44]:= ImposeSym[T[-f, -h, -l, -m] S[-a, -b, -c, -d],
          IndexList[-d, -f], Symmetric[{1, 2}]]

Out[44]= Sym[S T]
         (45)      abcdfhlm
```

If one would like to sort them alphabetically at any ther time one can use SortTensorsInSym

```
In[45]:= % /. {T -> S, S -> T}

Out[45]= Sym[T S]
         (45)      abcdfhlm

In[46]:= SortTensorsInSym@%

Out[46]= Sym[S T]
         (18)      fhlmabcd
```

If a tensor is inside the SymH object, but the symmetrization is not acting on its indices, one can move it outside the SymH object.

```
In[47]:= ImposeSym[S[-a, -b, -c, -d] T[-f, -h, -l, -m],
          IndexList[-a, -b], Symmetric[{1, 2}]]

Out[47]= Sym[S T]
         (12)      abcdfhlm

In[48]:= MoveTensorsOutsideSym@%

Out[48]= T      Sym[S]
          fhlm  (12)    abcd
```

One can also move tensors inside the SymH object.

```
In[49]:= MoveTensorsInsideSym@%

Out[49]= Sym[S T]
         (12)      abcdfhlm
```

Symmetrizations can be nested

```
In[50]:= ImposeSym[T[-a, -b, -c, -d], IndexList[-a, -b, -c, -d], Symmetric[{1, 2, 3}]]
```

Out[50]= $\underset{(123)}{\mathrm{Sym}}[T]_{abcd}$

```
In[51]:= ImposeSym[%, IndexList[-a, -b, -c, -d], Symmetric[{3, 4}]]
```

Out[51]= $\underset{(34)}{\mathrm{Sym}}\left[\underset{(123)}{\mathrm{Sym}}[T]\right]_{abcd}$

ExpandSym only expands the outermost SymH object

```
In[52]:= Expand@ExpandSym@%
```

Out[52]= $\frac{1}{2}\underset{(123)}{\mathrm{Sym}}[T]_{abcd} + \frac{1}{2}\underset{(123)}{\mathrm{Sym}}[T]_{abdc}$

```
In[53]:= Expand@ExpandSym@%
```

Out[53]= $\frac{1}{12}T_{abcd} + \frac{1}{12}T_{abdc} + \frac{1}{12}T_{acbd} + \frac{1}{12}T_{adbc} + \frac{1}{12}T_{bacd} + \frac{1}{12}T_{badc} + \frac{1}{12}T_{bcad} + \frac{1}{12}T_{bdac} + \frac{1}{12}T_{cabd} + \frac{1}{12}T_{cbad} + \frac{1}{12}T_{dabc} + \frac{1}{12}T_{dbac}$

Sometimes, when nesting symmeries the inner symmetrization might not be needed because of the outer symmetrization. Then this can be removed by RemoveSuperfluousInnerSym

```
In[54]:= ImposeSym[T[-a, -b, -c, -d], IndexList[-a, -b, -c, -d], Symmetric[{1, 2, 3}]]
```

Out[54]= $\underset{(123)}{\mathrm{Sym}}[T]_{abcd}$

```
In[55]:= ImposeSym[%, IndexList[-a, -b, -c, -d], Symmetric[{1, 2, 3, 4}]]
```

Out[55]= $\underset{(1234)}{\mathrm{Sym}}\left[\underset{(123)}{\mathrm{Sym}}[T]\right]_{abcd}$

```
In[56]:= RemoveSuperfluousInnerSym@%
```

Out[56]= $\underset{(1234)}{\mathrm{Sym}}[T]_{abcd}$

```
In[57]:= Expand[ExpandSym[%] - ExpandSym@ExpandSym[%%]]
```

Out[57]= 0

The functions for dealing with nested SymH objects work best when all tensors are innermost. To get that, use MoveTensorsInsideSym before you apply the second ImposeSym.

```
In[58]:= SymTSexpr1 = ImposeSym[T[-a, -b, -c, -d],
            IndexList[-a, -b, -c, -d], Symmetric[{1, 2, 3}]] S[-f, -h, -l, -m]
```

Out[58]= $S_{fhlm}\, \underset{(123)}{\text{Sym}}[T]_{abcd}$

```
In[59]:= ImposeSym[SymTSexpr1, IndexList[-a, -b, -c, -d], Symmetric[{1, 2, 3, 4}]]
```

Out[59]= $\underset{(5678)}{\text{Sym}}\left[S\, \underset{(123)}{\text{Sym}}[T]\right]_{fhlmabcd}$

```
In[60]:= RemoveSuperfluousInnerSym@%
```

Out[60]= $\underset{(5678)}{\text{Sym}}\left[S\, \underset{(123)}{\text{Sym}}[T]\right]_{fhlmabcd}$

```
In[61]:= MoveTensorsInsideSym@SymTSexpr1
```

Out[61]= $\underset{(567)}{\text{Sym}}[S\, T]_{fhlmabcd}$

```
In[62]:= ImposeSym[%, IndexList[-a, -b, -c, -d], Symmetric[{1, 2, 3, 4}]]
```

Out[62]= $\underset{(5678)}{\text{Sym}}\left[\underset{(567)}{\text{Sym}}[S\, T]\right]_{fhlmabcd}$

```
In[63]:= RemoveSuperfluousInnerSym@%
```

Out[63]= $\underset{(5678)}{\text{Sym}}[S\, T]_{fhlmabcd}$

```
In[64]:= MoveTensorsOutsideSym@%
```

Out[64]= $S_{fhlm}\, \underset{(1234)}{\text{Sym}}[T]_{abcd}$

Examples with a metric

Definition of a Lorentzian metric on the manifold:

```
In[65]:= DefMetric[{1, 3, 0}, g[-a, -b], CD]

** DefTensor: Defining symmetric metric tensor g[-a, -b].
** DefTensor: Defining antisymmetric tensor epsilong[-a, -b, -c, -d].
** DefTensor: Defining tensor Tetrag[-a, -b, -c, -d].
** DefTensor: Defining tensor Tetrag†[-a, -b, -c, -d].
** DefCovD: Defining covariant derivative CD[-a].
** DefTensor: Defining vanishing torsion tensor TorsionCD[a, -b, -c].
** DefTensor: Defining symmetric Christoffel tensor ChristoffelCD[a, -b, -c].
** DefTensor: Defining Riemann tensor RiemannCD[-a, -b, -c, -d].
** DefTensor: Defining symmetric Ricci tensor RicciCD[-a, -b].
** DefCovD:  Contractions of Riemann automatically replaced by Ricci.
** DefTensor: Defining Ricci scalar RicciScalarCD[].
** DefCovD:  Contractions of Ricci automatically replaced by RicciScalar.
** DefTensor: Defining symmetric Einstein tensor EinsteinCD[-a, -b].
** DefTensor: Defining Weyl tensor WeylCD[-a, -b, -c, -d].
** DefTensor: Defining symmetric TFRicci tensor TFRicciCD[-a, -b].
** DefTensor: Defining Kretschmann scalar KretschmannCD[].
** DefCovD:  Computing RiemannToWeylRules for dim 4
** DefCovD:  Computing RicciToTFRicci for dim 4
** DefCovD:  Computing RicciToEinsteinRules for dim 4
** DefTensor: Defining weight +2 density Detg[]. Determinant.
```

If you contract indices of a SymH object, this is means that the symmetrization is done first and then the contraction.

```
In[66]:= expr1a = ImposeSym[T[-a, -b, -c, -d], IndexList[-a, -b], Symmetric[{1, 2}]]
```

Out[66]= $\text{Sym}_{(12)}[T]_{abcd}$

```
In[67]:= expr1b = ContractMetric[expr1a * g[b, c]]
```

Out[67]= $\text{Sym}_{(12)}[T]_{ab}{}^{b}{}_{d}$

In[68]:= **ContractMetric[ExpandSym@expr1a * g[b, c]]**

Out[68]= $\frac{1}{2} T_{ab}{}^{b}{}_{d} + \frac{1}{2} T_{ba}{}^{b}{}_{d}$

In[69]:= **ExpandSym@expr1b**

Out[69]= $\frac{1}{2} \left(T_{ab}{}^{b}{}_{d} + T_{ba}{}^{b}{}_{d} \right)$

Observe that the symmetrization and trace operations do not commute in general as seen by this example:

In[70]:= **DefTensor[V[-a, -b, -c, -d], M4, StrongGenSet[{1, 2, 3, 4}, GenSet[-Cycles[{2, 3}]]]]**

** DefTensor: Defining tensor V[-a, -b, -c, -d].

In[71]:= **ContractMetric[V[-a, -b, -c, -d] g[b, c]]**

Out[71]= $V_{ab}{}^{b}{}_{d}$

In[72]:= **ToCanonical@%**

Out[72]= 0

In[73]:= **expr1a = ToCanonical@Symmetrize[%, {-a, -b}]**

Out[73]= 0

In[74]:= **Symmetrize[V[-a, -b, -c, -d], {-a, -b}]**

Out[74]= $\frac{1}{2} \left(V_{abcd} + V_{bacd} \right)$

In[75]:= **ToCanonical@%**

Out[75]= $\frac{1}{2} V_{abcd} + \frac{1}{2} V_{bacd}$

In[76]:= **expr1b = ToCanonical@ContractMetric[% g[b, c]]**

Out[76]= $\frac{1}{2} V^{b}{}_{abd}$

In[77]:= **ToCanonical[expr1a - expr1b]**

Out[77]= $-\frac{1}{2} V^{b}{}_{abd}$

The same example with SymH objects:

```
In[78]:= ContractMetric[V[-a, -b, -c, -d] g[b, c]]
```

Out[78]= $V_{ab}{}^{b}{}_{d}$

```
In[79]:= ToCanonical@%
```

Out[79]= 0

```
In[80]:= expr1a = ImposeSym[%, IndexList[-a, -b]]
```

Out[80]= 0

```
In[81]:= ImposeSym[V[-a, -b, -c, -d], IndexList[-a, -b]]
```

Out[81]= $\underset{(12)}{\mathrm{Sym}[V]}{}_{abcd}$

```
In[82]:= expr1b = ToCanonical@ContractMetric[% g[b, c]]
```

Out[82]= $\underset{(12)}{\mathrm{Sym}[V]}{}_{a}{}^{b}{}_{bd}$

```
In[83]:= ToCanonical@ExpandSym[expr1a - expr1b]
```

Out[83]= $-\frac{1}{2} V^{b}{}_{abd}$

However, symmetrization and traces do commute if the traces and symmetrizations are taken over disjoint sets of indices.

```
In[84]:= ContractMetric[V[-a, -b, -c, -d] g[b, d]]
```

Out[84]= $V_{abc}{}^{b}$

```
In[85]:= ToCanonical@%
```

Out[85]= $-V_{ac}{}^{b}{}_{b}$

```
In[86]:= expr1a = ImposeSym[%, IndexList[-a, -c]]
```

Out[86]= $-\underset{(12)}{\mathrm{Sym}[V]}{}_{ac}{}^{b}{}_{b}$

```
In[87]:= ImposeSym[V[-a, -b, -c, -d], IndexList[-a, -c]]
```

Out[87]= $\underset{(13)}{\mathrm{Sym}[V]}{}_{abcd}$

```
In[88]:= expr1b = ToCanonical@ContractMetric[% g[b, d]]
```

Out[88]= $\underset{(13)}{\mathrm{Sym}[V]}{}_{a}{}^{b}{}_{cb}$

In[89]:= `expr1b - expr1a`

Out[89]= $\underset{(12)}{\mathrm{Sym}[V]}{}_{ac}{}^{b}{}_{b} + \underset{(13)}{\mathrm{Sym}[V]}{}_{a}{}^{b}{}_{cb}$

In[90]:= `ExpandSym[expr1b - expr1a]`

Out[90]= $\frac{1}{2}\left(V_{ac}{}^{b}{}_{b} + V_{ca}{}^{b}{}_{b}\right) + \frac{1}{2}\left(V_{a}{}^{b}{}_{cb} + V_{c}{}^{b}{}_{ab}\right)$

In[91]:= `ToCanonical@%`

Out[91]= 0

In[92]:= `ToCanonical@CanonicalizeGroupInSym[expr1b - expr1a]`

Out[92]= 0

Sometimes there is a metric inside the SymH object that later can be contracted. The code for ContractMetricsInsideSym is ***new*** in release 0.8.3 and is not well tested.

In[93]:= `ImposeSym[V[-a, -b, -c, -d] g[-f, -h], IndexList[-a, -f]]`

Out[93]= $\underset{(13)}{\mathrm{Sym}[g\,V]}{}_{fhabcd}$

In[94]:= `ContractMetric[g[h, b] %]`

Out[94]= $\underset{(13)}{\mathrm{Sym}[g\,V]}{}_{fha}{}^{h}{}_{cd}$

In[95]:= `ContractMetricsInsideSym@%`

Out[95]= $\underset{(12)}{\mathrm{Sym}[V]}{}_{afcd}$

In[96]:= `ToCanonical[ExpandSym@% - ContractMetric@ExpandSym@%%]`

Out[96]= 0

Examples with a derivative

This far only derivatives compatible with the metric are handled. Prefix notation is used.

In[97]:= `CD[-a]@T[-b, -c, -d, -f]`

Out[97]= $\nabla_{a} T_{bcdf}$

The derivative index is the first index of the SymH object. That is, we keep the order of the indices when we use prefix notation.

In[98]:= `expr1a = ImposeSym[%, IndexList[-a, -b, -d]]`

Out[98]= $\underset{(124)}{\mathrm{Sym}[\nabla T]}{}_{abcdf}$

In many cases the TexAct package can produce the correct TeX code for the expression.

In[99]:= **Tex["▽"] = "\\nabla";**

In[100]:=

```
TexPrint@expr1a
```

Out[100]=

```
\nabla_{(a}T_{b|c|d)f}
```

The derivative is internally represented by CovarD[CD,T,{-TangentM4}]. The list {-TangentM4} gives VBundle for the derivative index.

In[101]:=

```
InputForm@expr1a
```

Out[101]//InputForm=

```
SymH[{CovarD[CD, T, {-TangentM4}]}, StrongGenSet[{1, 2, 4},
   GenSet[Cycles[{1, 2}], Cycles[{2, 4}]]], "(124)"][-a,
 -b, -c, -d, -f]
```

Higher order derivatives also work

In[102]:=

```
CD[-a]@CD[-b]@T[-c, -d, -f, -h]
```

Out[102]=

$\nabla_a \nabla_b T_{cdfh}$

In[103]:=

```
ImposeSym[%, IndexList[-a, -b, -c, -d]]
```

Out[103]=

$\underset{(1234)}{\text{Sym}}[\nabla \nabla T]_{abcdfh}$

In[104]:=

```
TexPrint@%
```

Out[104]=

```
\nabla_{(a}\nabla_{b}T_{cd)fh}
```

Examples with irreducible decompositions of Spinors *(new feature)*

The IrrDecompose code needs spinors and loads it if it is not already loaded

```
In[105]:=
    IrrDecompose[1]

        The Spinors package is needed for this.

        ------------------------------------------------------------

        Package xAct`Spinors`  version 1.0.3, {2012, 5, 5}

        CopyRight (C) 2006-2012, Alfonso Garcia-Parrado Gomez-Lobo
          and Jose M. Martin-Garcia, under the General Public License.

        ------------------------------------------------------------

        These packages come with ABSOLUTELY NO WARRANTY; for details type
          Disclaimer[]. This is free software, and you are welcome to redistribute
          it under certain conditions. See the General Public License for details.

        ------------------------------------------------------------

Out[105]=
    1
```

Define the spin structure

```
In[106]:=
    SetOptions[DefTensor, DefInfo → False]

Out[106]=
    {Dagger → Real, Master → Null, PrintAs → Identity,
     VanishingQ → False, ForceSymmetries → False, WeightOfTensor → 0,
     FrobeniusQ → False, OrthogonalTo → {}, ProjectedWith → {},
     ProtectNewSymbol :→ $ProtectNewSymbols, DefInfo → False, TensorID → {}}
```

In[107]:=
```
DefSpinStructure[g, Spin, {A, B, C, D, F, H, L, M, P, Q, R},
 ϵ, σ, CDe, {";", "∇"}, SpinorPrefix -> SP, SpinorMark -> "S"]
```

** DefVBundle: Defining vbundle Spin.

ValidateSymbol::capital : System name C is overloaded as an abstract index.

ValidateSymbol::capital : System name D is overloaded as an abstract index.

** DefVBundle: Defining conjugated vbundle Spin†
. Assuming fixed anti-isomorphism between Spin and Spin†

** DefCovD: Defining covariant derivative CDe[-a].

** DefTensor: Defining
nonsymmetric AChristoffel tensor AChristoffelCDe[A, -b, -C].

** DefTensor: Defining
nonsymmetric AChristoffel tensor AChristoffelCDe†[A†, -b, -C†].

** DefTensor: Defining FRiemann tensor
FRiemannCDe[-a, -b, -C, D]. Antisymmetric only in the first pair.

** DefTensor: Defining FRiemann tensor
FRiemannCDe†[-a, -b, -C†, D†]. Antisymmetric only in the first pair.

Define some spinors

In[108]:=
```
DefSpinor[T1[-A], M4]
```

In[109]:=
```
DefSpinor[T2[-A, -B], M4]
```

In[110]:=
```
DefSpinor[T3[-A, -B, -C], M4]
```

In[111]:=
```
DefSpinor[T4[-A, -B, -C, -D], M4]
```

In[112]:=
```
DefSpinor[S4[-A, -B, -C, -D], M4, Symmetric[{-A, -B, -C}]]
```

In[113]:=
```
DefSpinor[Ω4[-A, -B, -C, -D], M4, Symmetric[{-A, -B, -C, -D}], PrintAs -> "Ω"]
```

The basic example is

In[114]:=
```
IrrDecT2 = T2[-A, -B] == IrrDecompose[T2[-A, -B]]
```

Out[114]=

$$T2_{AB} == -\frac{1}{2} T2^{C}{}_{C}\ \epsilon_{AB} + \underset{(12)}{\mathrm{Sym}}[T2]_{AB}$$

It can be tested by making all possible contractions and symmetrizations

In[115]:=
```
Times[ϵ[A, B], #] & /@ IrrDecT2
```
Out[115]=

$$T2_{AB}\ \epsilon^{AB} == \epsilon^{AB}\left(-\frac{1}{2}\ T2^{C}{}_{C}\ \epsilon_{AB} + \underset{(12)}{\mathrm{Sym}}[T2]_{AB}\right)$$

In[116]:=
```
ToCanonical@ContractMetric@%
```
Out[116]=
```
True
```

In[117]:=
```
ImposeSym[#, IndexList[-A, -B]] & /@ IrrDecT2
```
Out[117]=
```
True
```

A sligthly more complicated case

In[118]:=
```
IrrDecT3 = T3[-A, -B, -C] == IrrDecompose[T3[-A, -B, -C]]
```
Out[118]=

$$T3_{ABC} == -\frac{1}{6}\ T3^{D}{}_{CD}\ \epsilon_{AB} - \frac{1}{6}\ T3^{D}{}_{DC}\ \epsilon_{AB} - \frac{1}{6}\ T3^{D}{}_{BD}\ \epsilon_{AC} - \frac{1}{6}\ T3^{D}{}_{DB}\ \epsilon_{AC} - \frac{1}{2}\ T3_{A}{}^{D}{}_{D}\ \epsilon_{BC} + \underset{(123)}{\mathrm{Sym}}[T3]_{ABC}$$

The IrrDecompose code uses the symmetries of the expression to simplify it

In[119]:=

```
IrrDecT4 = T4[-A, -B, -C, -D] == IrrDecompose[T4[-A, -B, -C, -D]]
```

Out[119]=

$$
\begin{aligned}
T4_{ABCD} == {} & \frac{1}{12} T4^{FH}{}_{FH}\, \epsilon_{AD}\, \epsilon_{BC} + \frac{1}{12} T4^{FH}{}_{HF}\, \epsilon_{AD}\, \epsilon_{BC} + \\
& \frac{1}{12} T4^{FH}{}_{FH}\, \epsilon_{AC}\, \epsilon_{BD} + \frac{1}{12} T4^{FH}{}_{HF}\, \epsilon_{AC}\, \epsilon_{BD} + \frac{1}{4} T4^{F}{}_{F}{}^{H}{}_{H}\, \epsilon_{AB}\, \epsilon_{CD} - \\
& \frac{1}{2} \epsilon_{CD}\, \underset{(12)}{\mathrm{Sym}}[T4]_{AB}{}^{F}{}_{F} - \frac{1}{6} \epsilon_{BD}\, \underset{(13)}{\mathrm{Sym}}[T4]_{A}{}^{F}{}_{CF} - \frac{1}{6} \epsilon_{BC}\, \underset{(13)}{\mathrm{Sym}}[T4]_{A}{}^{F}{}_{DF} - \\
& \frac{1}{6} \epsilon_{BD}\, \underset{(14)}{\mathrm{Sym}}[T4]_{A}{}^{F}{}_{FC} - \frac{1}{6} \epsilon_{BC}\, \underset{(14)}{\mathrm{Sym}}[T4]_{A}{}^{F}{}_{FD} - \frac{1}{12} \epsilon_{AD}\, \underset{(23)}{\mathrm{Sym}}[T4]^{F}{}_{BCF} - \\
& \frac{1}{12} \epsilon_{AC}\, \underset{(23)}{\mathrm{Sym}}[T4]^{F}{}_{BDF} - \frac{1}{12} \epsilon_{AB}\, \underset{(23)}{\mathrm{Sym}}[T4]^{F}{}_{CDF} - \frac{1}{12} \epsilon_{AD}\, \underset{(24)}{\mathrm{Sym}}[T4]^{F}{}_{BFC} - \\
& \frac{1}{12} \epsilon_{AC}\, \underset{(24)}{\mathrm{Sym}}[T4]^{F}{}_{BFD} - \frac{1}{12} \epsilon_{AB}\, \underset{(24)}{\mathrm{Sym}}[T4]^{F}{}_{CFD} - \frac{1}{12} \epsilon_{AD}\, \underset{(34)}{\mathrm{Sym}}[T4]^{F}{}_{FBC} - \\
& \frac{1}{12} \epsilon_{AC}\, \underset{(34)}{\mathrm{Sym}}[T4]^{F}{}_{FBD} - \frac{1}{12} \epsilon_{AB}\, \underset{(34)}{\mathrm{Sym}}[T4]^{F}{}_{FCD} + \underset{(1234)}{\mathrm{Sym}}[T4]_{ABCD}
\end{aligned}
$$

In[120]:=

```
IrrDecS4 = S4[-A, -B, -C, -D] == IrrDecompose[S4[-A, -B, -C, -D]]
```

Out[120]=

$$
\begin{aligned}
S4_{ABCD} == {} & -\frac{1}{12} S4_{CD}{}^{F}{}_{F}\, \epsilon_{AB} - \frac{1}{12} S4_{BD}{}^{F}{}_{F}\, \epsilon_{AC} - \frac{1}{12} S4_{BC}{}^{F}{}_{F}\, \epsilon_{AD} - \\
& \frac{1}{6} S4_{AD}{}^{F}{}_{F}\, \epsilon_{BC} - \frac{1}{6} S4_{AC}{}^{F}{}_{F}\, \epsilon_{BD} - \frac{1}{2} S4_{AB}{}^{F}{}_{F}\, \epsilon_{CD} + \underset{(1234)}{\mathrm{Sym}}[S4]_{ABCD}
\end{aligned}
$$

One can also simplify it afterwards

In[121]:=

```
ToCanonical@CanonicalizeGroupInSym@RemoveSuperfluousSym[IrrDecT4 /. T4 -> S4]
```

Out[121]=

$$
\begin{aligned}
S4_{ABCD} == {} & -\frac{1}{12} S4_{CD}{}^{F}{}_{F}\, \epsilon_{AB} - \frac{1}{12} S4_{BD}{}^{F}{}_{F}\, \epsilon_{AC} - \frac{1}{12} S4_{BC}{}^{F}{}_{F}\, \epsilon_{AD} - \\
& \frac{1}{6} S4_{AD}{}^{F}{}_{F}\, \epsilon_{BC} - \frac{1}{6} S4_{AC}{}^{F}{}_{F}\, \epsilon_{BD} - \frac{1}{2} S4_{AB}{}^{F}{}_{F}\, \epsilon_{CD} + \underset{(1234)}{\mathrm{Sym}}[S4]_{ABCD}
\end{aligned}
$$

In[122]:=

```
IrrDecS4[[2]] - %[[2]]
```

Out[122]=

```
0
```

The decomposition can be simplified further. We see that the left hand side of IrrDecS4 is symmetric in {-A,-B,-C}, but this is not obvoius on the right hand side.

In[123]:=

```
IrrDecS4
```

Out[123]=

$$S4_{ABCD} == -\frac{1}{12} S4_{CD}{}^{F}{}_{F}\ \epsilon_{AB} - \frac{1}{12} S4_{BD}{}^{F}{}_{F}\ \epsilon_{AC} - \frac{1}{12} S4_{BC}{}^{F}{}_{F}\ \epsilon_{AD} - \frac{1}{6} S4_{AD}{}^{F}{}_{F}\ \epsilon_{BC} - \frac{1}{6} S4_{AC}{}^{F}{}_{F}\ \epsilon_{BD} - \frac{1}{2} S4_{AB}{}^{F}{}_{F}\ \epsilon_{CD} + \underset{(1234)}{\text{Sym}}[S4]_{ABCD}$$

In[124]:=

```
IrrDecS4[[1]] == ToCanonical@
  ImposeSymmetry[IrrDecS4[[2]], IndexList[-A, -B, -C], Symmetric[{1, 2, 3}]]
```

Out[124]=

$$S4_{ABCD} == -\frac{1}{4} S4_{BC}{}^{F}{}_{F}\ \epsilon_{AD} - \frac{1}{4} S4_{AC}{}^{F}{}_{F}\ \epsilon_{BD} - \frac{1}{4} S4_{AB}{}^{F}{}_{F}\ \epsilon_{CD} + \underset{(1234)}{\text{Sym}}[S4]_{ABCD}$$

The function CompleteIrrDecompose automatically finds this symmetry and imposes it

In[125]:=

```
IrrDecS4b =
 S4[-A, -B, -C, -D] == CompleteIrrDecompose[S4[-A, -B, -C, -D], TimeVerbose -> True]
```

```
IrrDecompose timing:0.1092001

CompleteIrrDecompose timing:0.0624002
```

Out[125]=

$$S4_{ABCD} == -\frac{1}{4} S4_{BC}{}^{F}{}_{F}\ \epsilon_{AD} - \frac{1}{4} S4_{AC}{}^{F}{}_{F}\ \epsilon_{BD} - \frac{1}{4} S4_{AB}{}^{F}{}_{F}\ \epsilon_{CD} + \underset{(1234)}{\text{Sym}}[S4]_{ABCD}$$

Larger cases can be handled

In[126]:=

```
Ω4[-A, -B, -C, L] Ω4[-D, -F, -H, -L] ==
 CompleteIrrDecompose[Ω4[-A, -B, -C, L] Ω4[-D, -F, -H, -L], TimeVerbose -> True]
```

```
IrrDecompose timing:1.8720033
```

```
CompleteIrrDecompose timing:0.5460010
```

Out[126]=

$$\Omega_{ABC}{}^{L}\,\Omega_{DFHL} == -\frac{1}{24}\epsilon_{AH}\,\epsilon_{BF}\,\epsilon_{CD}\,\Omega_{LMPQ}\,\Omega^{LMPQ} - \frac{1}{24}\epsilon_{AF}\,\epsilon_{BH}\,\epsilon_{CD}\,\Omega_{LMPQ}\,\Omega^{LMPQ} - \frac{1}{24}\epsilon_{AH}\,\epsilon_{BD}\,\epsilon_{CF}\,\Omega_{LMPQ}\,\Omega^{LMPQ} - \frac{1}{24}\epsilon_{AD}\,\epsilon_{BH}\,\epsilon_{CF}\,\Omega_{LMPQ}\,\Omega^{LMPQ} - \frac{1}{24}\epsilon_{AF}\,\epsilon_{BD}\,\epsilon_{CH}\,\Omega_{LMPQ}\,\Omega^{LMPQ} - \frac{1}{24}\epsilon_{AD}\,\epsilon_{BF}\,\epsilon_{CH}\,\Omega_{LMPQ}\,\Omega^{LMPQ} - \frac{1}{6}\epsilon_{CH}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{LM}{}_{DFLM} - \frac{1}{6}\epsilon_{CF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{LM}{}_{DHLM} - \frac{1}{6}\epsilon_{CD}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{LM}{}_{FHLM} - \frac{1}{6}\epsilon_{BH}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{LM}{}_{DFLM} - \frac{1}{6}\epsilon_{BF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{LM}{}_{DHLM} - \frac{1}{6}\epsilon_{BD}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{LM}{}_{FHLM} - \frac{1}{6}\epsilon_{AH}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{LM}{}_{DFLM} - \frac{1}{6}\epsilon_{AF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{LM}{}_{DHLM} - \frac{1}{6}\epsilon_{AD}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{LM}{}_{FHLM}$$

And even larger

In[127]:=

```
Ω4[-A, -B, -C, -D] Ω4[-F, -H, -L, -M] ==
 CompleteIrrDecompose[Ω4[-A, -B, -C, -D] Ω4[-F, -H, -L, -M], TimeVerbose -> True]
```

```
IrrDecompose timing:9.3756165
```

```
CompleteIrrDecompose timing:47.0808827
```

Out[127]=

$$\Omega_{ABCD}\,\Omega_{FHLM} == \frac{1}{120}\epsilon_{AM}\,\epsilon_{BL}\,\epsilon_{CH}\,\epsilon_{DF}\,\Omega_{PQRR1}\,\Omega^{PQRR1} + \frac{1}{120}\epsilon_{AL}\,\epsilon_{BM}\,\epsilon_{CH}\,\epsilon_{DF}\,\Omega_{PQRR1}\,\Omega^{PQRR1} + \frac{1}{120}\epsilon_{AM}\,\epsilon_{BH}\,\epsilon_{CL}\,\epsilon_{DF}\,\Omega_{PQRR1}\,\Omega^{PQRR1} + \frac{1}{120}\epsilon_{AH}\,\epsilon_{BM}\,\epsilon_{CL}\,\epsilon_{DF}\,\Omega_{PQRR1}\,\Omega^{PQRR1} + \frac{1}{120}\epsilon_{AL}\,\epsilon_{BH}\,\epsilon_{CM}\,\epsilon_{DF}\,\Omega_{PQRR1}\,\Omega^{PQRR1} +$$

$$\frac{1}{120}\epsilon_{AH}\epsilon_{BL}\epsilon_{CM}\epsilon_{DF}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AM}\epsilon_{BL}\epsilon_{CF}\epsilon_{DH}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AL}\epsilon_{BM}\epsilon_{CF}\epsilon_{DH}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AM}\epsilon_{BF}\epsilon_{CL}\epsilon_{DH}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AF}\epsilon_{BM}\epsilon_{CL}\epsilon_{DH}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AL}\epsilon_{BF}\epsilon_{CM}\epsilon_{DH}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AF}\epsilon_{BL}\epsilon_{CM}\epsilon_{DH}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AM}\epsilon_{BH}\epsilon_{CF}\epsilon_{DL}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AH}\epsilon_{BM}\epsilon_{CF}\epsilon_{DL}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AM}\epsilon_{BF}\epsilon_{CH}\epsilon_{DL}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AF}\epsilon_{BM}\epsilon_{CH}\epsilon_{DL}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AH}\epsilon_{BF}\epsilon_{CM}\epsilon_{DL}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AF}\epsilon_{BH}\epsilon_{CM}\epsilon_{DL}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AL}\epsilon_{BH}\epsilon_{CF}\epsilon_{DM}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AH}\epsilon_{BL}\epsilon_{CF}\epsilon_{DM}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AL}\epsilon_{BF}\epsilon_{CH}\epsilon_{DM}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AF}\epsilon_{BL}\epsilon_{CH}\epsilon_{DM}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AH}\epsilon_{BF}\epsilon_{CL}\epsilon_{DM}\Omega_{PQRR1}\Omega^{PQRR1}+$$
$$\frac{1}{120}\epsilon_{AF}\epsilon_{BH}\epsilon_{CL}\epsilon_{DM}\Omega_{PQRR1}\Omega^{PQRR1}+\frac{1}{42}\epsilon_{CM}\epsilon_{DL}\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{FHPQ}+$$
$$\frac{1}{42}\epsilon_{CL}\epsilon_{DM}\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{FHPQ}+\frac{1}{42}\epsilon_{CM}\epsilon_{DH}\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{FLPQ}+$$
$$\frac{1}{42}\epsilon_{CH}\epsilon_{DM}\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{FLPQ}+\frac{1}{42}\epsilon_{CL}\epsilon_{DH}\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{FMPQ}+$$
$$\frac{1}{42}\epsilon_{CH}\epsilon_{DL}\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{FMPQ}+\frac{1}{42}\epsilon_{CM}\epsilon_{DF}\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{HLPQ}+$$

$$\frac{1}{42} \epsilon_{CF} \epsilon_{DM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{HLPQ} + \frac{1}{42} \epsilon_{CL} \epsilon_{DF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{HMPQ} +$$
$$\frac{1}{42} \epsilon_{CF} \epsilon_{DL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{HMPQ} + \frac{1}{42} \epsilon_{CH} \epsilon_{DF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{LMPQ} +$$
$$\frac{1}{42} \epsilon_{CF} \epsilon_{DH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AB}{}^{PQ}{}_{LMPQ} + \frac{1}{42} \epsilon_{BM} \epsilon_{DL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{FHPQ} +$$
$$\frac{1}{42} \epsilon_{BL} \epsilon_{DM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{FHPQ} + \frac{1}{42} \epsilon_{BM} \epsilon_{DH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{FLPQ} +$$
$$\frac{1}{42} \epsilon_{BH} \epsilon_{DM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{FLPQ} + \frac{1}{42} \epsilon_{BL} \epsilon_{DH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{FMPQ} +$$
$$\frac{1}{42} \epsilon_{BH} \epsilon_{DL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{FMPQ} + \frac{1}{42} \epsilon_{BM} \epsilon_{DF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{HLPQ} +$$
$$\frac{1}{42} \epsilon_{BF} \epsilon_{DM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{HLPQ} + \frac{1}{42} \epsilon_{BL} \epsilon_{DF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{HMPQ} +$$
$$\frac{1}{42} \epsilon_{BF} \epsilon_{DL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{HMPQ} + \frac{1}{42} \epsilon_{BH} \epsilon_{DF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{LMPQ} +$$
$$\frac{1}{42} \epsilon_{BF} \epsilon_{DH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AC}{}^{PQ}{}_{LMPQ} + \frac{1}{42} \epsilon_{BM} \epsilon_{CL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{FHPQ} +$$
$$\frac{1}{42} \epsilon_{BL} \epsilon_{CM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{FHPQ} + \frac{1}{42} \epsilon_{BM} \epsilon_{CH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{FLPQ} +$$
$$\frac{1}{42} \epsilon_{BH} \epsilon_{CM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{FLPQ} + \frac{1}{42} \epsilon_{BL} \epsilon_{CH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{FMPQ} +$$
$$\frac{1}{42} \epsilon_{BH} \epsilon_{CL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{FMPQ} + \frac{1}{42} \epsilon_{BM} \epsilon_{CF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{HLPQ} +$$
$$\frac{1}{42} \epsilon_{BF} \epsilon_{CM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{HLPQ} + \frac{1}{42} \epsilon_{BL} \epsilon_{CF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{HMPQ} +$$
$$\frac{1}{42} \epsilon_{BF} \epsilon_{CL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{HMPQ} + \frac{1}{42} \epsilon_{BH} \epsilon_{CF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{LMPQ} +$$
$$\frac{1}{42} \epsilon_{BF} \epsilon_{CH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{AD}{}^{PQ}{}_{LMPQ} + \frac{1}{42} \epsilon_{AM} \epsilon_{DL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{FHPQ} +$$
$$\frac{1}{42} \epsilon_{AL} \epsilon_{DM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{FHPQ} + \frac{1}{42} \epsilon_{AM} \epsilon_{DH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{FLPQ} +$$
$$\frac{1}{42} \epsilon_{AH} \epsilon_{DM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{FLPQ} + \frac{1}{42} \epsilon_{AL} \epsilon_{DH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{FMPQ} +$$
$$\frac{1}{42} \epsilon_{AH} \epsilon_{DL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{FMPQ} + \frac{1}{42} \epsilon_{AM} \epsilon_{DF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{HLPQ} +$$
$$\frac{1}{42} \epsilon_{AF} \epsilon_{DM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{HLPQ} + \frac{1}{42} \epsilon_{AL} \epsilon_{DF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{HMPQ} +$$
$$\frac{1}{42} \epsilon_{AF} \epsilon_{DL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{HMPQ} + \frac{1}{42} \epsilon_{AH} \epsilon_{DF} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{LMPQ} +$$
$$\frac{1}{42} \epsilon_{AF} \epsilon_{DH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BC}{}^{PQ}{}_{LMPQ} + \frac{1}{42} \epsilon_{AM} \epsilon_{CL} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{FHPQ} +$$
$$\frac{1}{42} \epsilon_{AL} \epsilon_{CM} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{FHPQ} + \frac{1}{42} \epsilon_{AM} \epsilon_{CH} \underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{FLPQ} +$$

$$\frac{1}{42}\epsilon_{AH}\epsilon_{CM}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{FLPQ}+\frac{1}{42}\epsilon_{AL}\epsilon_{CH}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{FMPQ}+$$
$$\frac{1}{42}\epsilon_{AH}\epsilon_{CL}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{FMPQ}+\frac{1}{42}\epsilon_{AM}\epsilon_{CF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{HLPQ}+$$
$$\frac{1}{42}\epsilon_{AF}\epsilon_{CM}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{HLPQ}+\frac{1}{42}\epsilon_{AL}\epsilon_{CF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{HMPQ}+$$
$$\frac{1}{42}\epsilon_{AF}\epsilon_{CL}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{HMPQ}+\frac{1}{42}\epsilon_{AH}\epsilon_{CF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{LMPQ}+$$
$$\frac{1}{42}\epsilon_{AF}\epsilon_{CH}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{BD}{}^{PQ}{}_{LMPQ}+\frac{1}{42}\epsilon_{AM}\epsilon_{BL}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{FHPQ}+$$
$$\frac{1}{42}\epsilon_{AL}\epsilon_{BM}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{FHPQ}+\frac{1}{42}\epsilon_{AM}\epsilon_{BH}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{FLPQ}+$$
$$\frac{1}{42}\epsilon_{AH}\epsilon_{BM}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{FLPQ}+\frac{1}{42}\epsilon_{AL}\epsilon_{BH}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{FMPQ}+$$
$$\frac{1}{42}\epsilon_{AH}\epsilon_{BL}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{FMPQ}+\frac{1}{42}\epsilon_{AM}\epsilon_{BF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{HLPQ}+$$
$$\frac{1}{42}\epsilon_{AF}\epsilon_{BM}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{HLPQ}+\frac{1}{42}\epsilon_{AL}\epsilon_{BF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{HMPQ}+$$
$$\frac{1}{42}\epsilon_{AF}\epsilon_{BL}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{HMPQ}+\frac{1}{42}\epsilon_{AH}\epsilon_{BF}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{LMPQ}+$$
$$\frac{1}{42}\epsilon_{AF}\epsilon_{BH}\,\underset{(1256)}{\mathrm{Sym}}[\Omega\,\Omega]_{CD}{}^{PQ}{}_{LMPQ}+\underset{(12\,345\,678)}{\mathrm{Sym}}[\Omega\,\Omega]_{ABCDFHLM}$$

Notes

In[128]:=

```
MaxMemoryUsed[]
```

Out[128]=

```
109 220 056
```

In[129]:=

```
TimeUsed[]
```

Out[129]=

```
58.546
```

Note: For further information about `SymManipulator``, and to be kept informed about new releases, you may contact the author electronically at thomas.backdahl@aei.mpg.de. This is `SymManipulatorDoc.nb`, the docfile of `SymManipulator``, currently in version 0.8.3.

In[130]:=
```
?xAct`SymManipulator`*
```

▼ xAct`SymManipulator`

CanonicalizeGroupInSym	ImposeSuperfluousSym	RemoveTrivialSym
CompatibleSymmetric	ImposeSym	SmartAntisymmetrize
CompleteIrrDecImposeNewMethod	InertHeadHead	SmartExpand
CompleteIrrDecompose	InternalCommutingSymmetry	SmartSymmetrize
ContractMetricsInsideSym	IrrDecompose	SortTensorsInSym
CovarD	MoveSymIndicesDown	SubgroupQ
deltaH	MoveTensorsInsideSym	SymH
Disclaimer	MoveTensorsOutsideSym	TensorToZeroRule
ExpandSym	RemoveSuperfluousInnerSym	ZeroTensor
ExpandSymOneIndex	RemoveSuperfluousSym	$Version
ImposeLargerSym	RemoveSym	